TECHNOLOGY



Calculating Loan Amortization

If you're planning on financing a capital expenditure, you can use the PMT, IPMT, and PPMT functions in Excel to quickly develop an amortization table. Excel uses the PMT function to calculate loan payments, while the PPMT and IPMT functions will calculate the portions of that payment attributable to principal and interest for each loan payment.

The Input Variables

Near the top of your worksheet, enter labels and create cells to hold the principal, term, and interest rate (see Figure 1). For this example, I will use a loan of \$1 million financed over 60 months at an interest rate of 3.75%. Enter 1000000 in cell B2, 60 in cell B3, and 3.75% in cell B4.

Calculate the Monthly Payment

The PMT function requires three arguments to calculate the loan payment: Rate, NPer, and PV.

Rate is the interest rate per period for the loan. While banks will quote you an annual interest rate, you will usually calculate monthly payments for the loan. This is one of the more confusing

	C9	• (*		fx =IPMT(\$B\$4/12,\$A9,\$B\$3,-\$B\$2)	
4	A	В	С	D	E	F
1						
2	Loan	\$1,000,000				
3	Term	60				
4	Int Rate	3.75%				
5	Payment	\$18,303.92				
6						
7	Period	Month	Interest	Principal	Balance	
8		4/1/2013			\$1,000,000.00	
9	1	5/1/2013	\$3,125.00	\$15,178.92	\$984,821.08	
10	2	6/1/2013	\$3,077.57	\$15,226.35	\$969,594.73	
11	3	7/1/2013	\$3,029.98	\$15,273.93	\$954,320.79	

aspects of the formula. The 3.75% in cell B4 is an annual interest rate. Any time you enter the interest rate in the formula, you'll need to use B4/12 for monthly payments or B4/4 for quarterly payments.

NPer is the number of periods during the loan. This is the number stored in cell B3.

PV is the amount borrowed, also known as the present value. If you want to show the monthly payments as positive numbers, use a negative number for present value. For example, we'll use -B2 for the PV in our formulas.

The formula in cell B5, =PMT(B4/12, B3,-B2), will calculate the monthly payment. The result is \$18,303.92.

Build a Monthly Amortization Table

Now build an amortization table so that you can see how the payments are applied to the loan each month, the remaining balance, and how much interest has been paid over the life of the loan. In row 7, enter headings of Period, Month (or Date), Interest, Principal, and Balance, as shown in Figure 1. In cell B8, enter the loan date. In cell E8, enter the formula =B2 to get the starting balance.

In cell A9, enter =ROW(A1). This formula will result in the number 1. The advantage of using the ROW function is that the formula will change to =ROW (A2), =ROW(A3), and so on as you copy the formula down, creating a series of period numbers, e.g., 1, 2, 3, and so forth.

In cell B9, enter =DATE(YEAR(B\$8), MONTH(B\$8)+A9,DAY(B\$8)). This will give you the date the loan payment is due.

In cell C9, calculate the interest amount for the first loan payment. The IPMT function is similar to the PMT function, but it includes a period number as the second argument: =IPMT(rate, per, nper, pv). The period number is stored in column A. Use =IPMT(\$B\$4/12,\$A9, \$B\$3,-\$B\$2). Note that the cell references to B4, B3, and B2 include dollar signs to make those references absolute. As you copy the formula in row 9 down the worksheet, they will continue to point to the same cells.

In cell D9, calculate the amount of the monthly payment applied to the principal balance. There are two ways to do this. The first is to use the PPMT function: =PPMT(\$B\$4/12,\$A9,\$B\$3,-\$B\$2). If you entered the period argument in cell C9 using a dollar sign for the column reference (\$A9), you can simply copy cell C9 to cell D9 and then change IPMT to PPMT. The second method is to calculate the principal payment as the payment in cell \$B\$5 minus the interest payment in this row: =\$B\$5-C9.

In cell E9, calculate the principal balance by using =E8-D9. This is the previous loan balance minus this month's principal payment.

The formulas in row 9 are designed to be copied down for each period. For a 60-period loan, copy A9:E9 down to the next 59 rows. You should end up with a final calculation in row 68 showing a loan balance very close to zero.

What-If Analysis

Once you've designed the worksheet, you can plug in new numbers for the loan amount in cell B2 or the interest rate in cell B4, and the worksheet will recalculate. To change the length of the loan or number of periods in cell B3, however, you'll have to adjust the number of rows in the amortization table. You can remember to do this manually each time you change the term, or you can adjust the formulas to do this automatically.

On the one hand, it's easy to adjust the worksheet manually. A change from 60 months to 120 months requires you to copy A68:E68 down another 60 rows. A change from 60 months to 48 months means you would need to clear rows 57 through 68.

On the other hand, it only takes a few minutes to change the formulas to appear automatically in additional rows based on the number entered in the period field in cell B3. You can then reuse this amortization worksheet as a template for all future loans.

Currently, the formula in A9 is =ROW(A1). Change this formula to =IF(ROW(A1)>\$B\$3," ",ROW(A1).

Edit the other four formulas in row 9. Add =IF(ISNUMBER(A9), at the beginning of each formula and ,"") at the end. For example, the formula in cell E9 would become =IF(ISNUMBER(A9),E8-D9,""). Then copy the formulas down the worksheet—make sure to copy at least as far down as the longest term you anticipate testing. As you change the term in cell B3, the results beyond the end of the loan will change to blank cells. **SF**

Bill Jelen is the host of MrExcel.com and the author of Learn Excel 2007-2010 from MrExcel. *Send questions for future articles to IMA@MrExcel.com.*